Pensieve header: A poly-time program to compute the ${\mathfrak g}_0$ invariant.

Reminders

1. Fill out “Graduate Course Participation Form” (see http://drorbn.net/AcademicPensieve/Classes/17-1350-AKT/GCP.jpg, in time).
2. Make sure that you have Mathematica and that you play with these programs!
3. Change meeting time?

Representing $\mathfrak{g}_0 = \langle h, e, l, f\rangle / ([e, l] = -e,\ [f, l] = f,\ [e, f] = h,\ [h, *] = 0)$

```
ρh = {{0, 0, -1}, {0, 0, 0}, {0, 0, 0}}; ρe = {{0, 0, 0}, {0, 0, 1}, {0, 0, 0}}; ρl = {{0, 0, 0}, {0, 1, 0}, {0, 0, 0}}; ρf = {{0, 1, 0}, {0, 0, 0}, {0, 0, 0}}; ρ0 = {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}};
B[x_?MatrixQ, y_?MatrixQ] := x.y - y.x;
```

```
{B[ρe, ρl] == -ρe, B[ρf, ρl] == ρf, B[ρe, ρf] == ρh, B[ρh, ρe] == ρ0, B[ρh, ρl] == ρ0, B[ρh, ρf] == ρ0}
{True, True, True, True, True, True}
```

Implementing $\mathfrak{g}_0$

```
PBWRule = {e → 1, l → 2, f → 3};
B[U@e, U@l] = -U@e; B[U@f, U@l] = U@f; B[U@e, U@f] = h U[];
```

```
$TD = 3; ħ /: ħ^d_. /; d > $TD := 0;
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule]; x_ < y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
U_i_[ℰ_] := ℰ /. {h → h_i, t → t_i, u_U :→ Replace[u, x_ :→ x_i, 1]};
B[U[(x_)_i_], U[(y_)_i_]] := B[U[x_i], U[y_i]] = U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, x_] = 0;
B[U[y_], U[x_]] := B[U[y], U[x]] = Simp[-B[U[x], U[y]]];
B[x_, y_] := x ** y - y ** x;
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

```
U[xx___, x_] ** U[y_, yy___] := If[x ≤ y, U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
```

```
UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[];
UU[l_, r___] := U[l] ** UU[r];
```

```
UProducts[{}, 0] = {UU[]};
UProducts[{}, n_Integer] /; n > 0 = {};
UProducts[{x_, xs___}, n_Integer] :=
  Sort@Flatten@Table[UU[x^k] ** u, {k, 0, n}, {u, UProducts[{xs}, n - k]}];
UProducts[xs_List, k_Integer, n_Integer] := UProducts[Flatten@Table[x_j, {x, xs}, {j, k}], n];
UProducts[any__, {n_}] := Flatten@Table[UProducts[any, k], {k, 0, n}];
```

```
r_{i_,j_} := Simp[ℏ (h_i UU[l_j] + UU[e_i, f_j])]
```

```
UExp[u_] := Module[{s, t, k},
   s = t = U[]; k = 0;
   While[k < 20 ∧ 0 =!= (t = t ** u), s += t / (++k) !];
   Simp[s]];
R_{i_,j_} := UExp[r_{i,j}];
```

```
m[i_, j_, k_][ℰ_] := Simp[ℰ /. {
    u_U :> UU @@ Join[DeleteCases[u, x_{i|j}], U @@ Cases[u, x_i_ :> x_k], U @@ Cases[u, x_j_ :> x_k]],
    h_{i|j} → h_k}]
```

Ordering Symbols; The Invariant of the Trefoil

Theorem. $R = e^{h\otimes l + e\otimes f} = \mathbb{O}\left(\exp\left(\mathrm{hl} + \frac{e^h-1}{h}\,\mathrm{ef}\;\middle|\; e\otimes \mathrm{lf}\right)\right)$.

Gentle Proof, Presented Brutely.

```
MatrixForm /@ {MatrixExp[h ρl + e ρf], MatrixExp[h ρl].MatrixExp[(e^h - 1)/h e ρf]}
```

$$\left\{\begin{pmatrix} 1 & \frac{e\,(-1+e^h)}{h} & 0 \\ 0 & e^h & 0 \\ 0 & 0 & 1\end{pmatrix}, \begin{pmatrix} 1 & \frac{e\,(-1+e^h)}{h} & 0 \\ 0 & e^h & 0 \\ 0 & 0 & 1\end{pmatrix}\right\}$$

```
O[poly_, specs___] := Module[{vs, us, z},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) :> (l /. x_i_ :> x_s));
  Simp@Total[CoefficientRules[Normal@Series[poly, {ℏ, 0, $TD}], vs] /. (p_ → c_) :> c UU @@ (us^p)]
 ]
```

Debt. How does this work?

```
Timing[$TD = 3; T1$1 = O[Exp[ℏ h l_1 + (e^{ℏ h} - 1)/h e_4 f_1 + ℏ h l_5 + (e^{ℏ h} - 1)/h e_2 f_5 + ℏ h l_3 + (e^{ℏ h} - 1)/h e_6 f_3],
    {l_1, f_1, e_2, l_3, f_3, e_4, l_5, f_5, e_6} → 1] /. h_1 → h]
```

$$\Big\{0.078125,\ \left(1 - 2\,h\,\hbar + h^2\,\hbar^2 + \frac{2\,h^3\,\hbar^3}{3}\right)\mathrm{U}[\,] + \left(3\,h\,\hbar - 6\,h^2\,\hbar^2 + 3\,h^3\,\hbar^3\right)\mathrm{U}[l_1] + \left(3\,\hbar - \frac{3\,h\,\hbar^2}{2} - \frac{3\,h^2\,\hbar^3}{2}\right)\mathrm{U}[e_1, f_1] + \left(\frac{9\,h^2\,\hbar^2}{2} - 9\,h^3\,\hbar^3\right)\mathrm{U}[l_1, l_1] +$$
$$\left(9\,h\,\hbar^2 - \frac{9\,h^2\,\hbar^3}{2}\right)\mathrm{U}[e_1, l_1, f_1] + \frac{9}{2}\,h^3\,\hbar^3\,\mathrm{U}[l_1, l_1, l_1] + \left(\frac{9\,\hbar^2}{2} + \frac{9\,h\,\hbar^3}{2}\right)\mathrm{U}[e_1, e_1, f_1, f_1] +$$
$$\frac{27}{2}\,h^2\,\hbar^3\,\mathrm{U}[e_1, l_1, l_1, f_1] + \frac{27}{2}\,h\,\hbar^3\,\mathrm{U}[e_1, e_1, l_1, f_1, f_1] + \frac{9}{2}\,\hbar^3\,\mathrm{U}[e_1, e_1, e_1, f_1, f_1, f_1]\Big\}$$

The Big $\mathfrak{g}_0$ Lemma

1. $\mathbb{O}\left(e^{\gamma l + \beta e} \mid \mathrm{le}\right) = \mathbb{O}\left(e^{\gamma l + e^{\gamma}\beta e} \mid \mathrm{el}\right)$.
2. $\mathbb{O}\left(e^{\gamma l + \beta f} \mid \mathrm{fl}\right) = \mathbb{O}\left(e^{\gamma l + e^{\gamma}\beta f} \mid \mathrm{lf}\right)$.

3. $\mathbb{O}\left(e^{\beta e+\alpha f+\delta ef} \mid fe\right) = \mathbb{O}\left(\nu e^{\nu(-\alpha\beta h+\beta e+\alpha f+\delta ef)} \mid ef\right)$, with $\nu = (1+h\delta)^{-1}$.

Gentle Proofs of 1 & 2 and of 3 at δ=0.

```
MatrixForm /@ {MatrixExp[γ ρl].MatrixExp[β ρe], MatrixExp[e^γ β ρe].MatrixExp[γ ρl]}
```

$$\left\{\begin{pmatrix}1&0&0\\0&e^\gamma&e^\gamma\beta\\0&0&1\end{pmatrix},\begin{pmatrix}1&0&0\\0&e^\gamma&e^\gamma\beta\\0&0&1\end{pmatrix}\right\}$$

```
MatrixForm /@ {MatrixExp[β ρf].MatrixExp[γ ρl], MatrixExp[γ ρl].MatrixExp[e^γ β ρf]}
```

$$\left\{\begin{pmatrix}1&e^\gamma\beta&0\\0&e^\gamma&0\\0&0&1\end{pmatrix},\begin{pmatrix}1&e^\gamma\beta&0\\0&e^\gamma&0\\0&0&1\end{pmatrix}\right\}$$

```
MatrixForm /@ {MatrixExp[α ρf].MatrixExp[β ρe], MatrixExp[-α β ρh].MatrixExp[β ρe].MatrixExp[α ρf]}
```

$$\left\{\begin{pmatrix}1&\alpha&\alpha\beta\\0&1&\beta\\0&0&1\end{pmatrix},\begin{pmatrix}1&\alpha&\alpha\beta\\0&1&\beta\\0&0&1\end{pmatrix}\right\}$$

Debt. Prove the full part 3!

The Main $\mathfrak{g}_0$ Theorem.

Raw Version. The $\mathfrak{g}_0$ invariant of any S-component tangle T can be written in the form $Z(T) = \mathbb{O}\left(\omega e^{L+Q} \mid \prod_{i\in S} e_i\, l_i\, f_i\right)$, where ω is a scalar (meaning, a rational function in the variables h_i and their exponentials $t_i = e^{h_i}$), where $L = \sum a_{ij}\, h_i\, l_j$ is a balanced quadratic in the variables h_i and l_j with integer coefficients a_{ij} and where $Q = \sum b_{ij}\, e_i\, f_j$ is a balanced quadratic in the variables e_i and f_j with scalar coefficients b_{ij}. Furthermore, after setting $h_i = h$ and $t_i = t$ for all i, the invariant $Z(T)$ is poly-time computable.

Proof. Indeed,

0. $R = e^{h\otimes l+e\otimes f} = \mathbb{O}\left(\exp\left(hl + \frac{e^h-1}{h} ef\right) \mid e\otimes lf\right)$,

1. $\mathbb{O}\left(e^{\gamma l+\beta e} \mid le\right) = \mathbb{O}\left(e^{\gamma l+e^\gamma \beta e} \mid el\right)$,

2. $\mathbb{O}\left(e^{\gamma l+\beta f} \mid fl\right) = \mathbb{O}\left(e^{\gamma l+e^\gamma \beta f} \mid lf\right)$,

3. $\mathbb{O}\left(e^{\beta e+\alpha f+\delta ef} \mid fe\right) = \mathbb{O}\left(\nu e^{\nu(-\alpha\beta h+\beta e+\alpha f+\delta ef)} \mid ef\right)$, with $\nu = (1+h\delta)^{-1}$,

and the rest is straight-forward.

old above / new below

```
CF[𝔼[ω_, L_, Q_]] := 𝔼[Simplify[ω], Simplify[L], Simplify[Q]];
𝔼 /: 𝔼[ω1_, L1_, Q1_] 𝔼[ω2_, L2_, Q2_] := CF@𝔼[ω1 ω2, L1 + L2, Q1 + Q2];
𝔼[ω1_, L1_, Q1_] ≡ 𝔼[ω2_, L2_, Q2_] := Simplify[ω1 == ω2 ∧ L1 == L2 ∧ Q1 == Q2];
```

0. $R = e^{h\otimes l+e\otimes f} = \mathbb{O}\left(\exp\left(hl + \frac{e^h-1}{h} ef\right) \mid e\otimes lf\right)$:

```
𝔼[X⁺_{i_,j_}] := 𝔼[1, h_i l_j, h_i^-1 (e^(h_i) - 1) e_i f_j];
𝔼[X⁻_{i_,j_}] := 𝔼[1, -h_i l_j, h_i^-1 (e^(-h_i) - 1) e_i f_j];
𝔼[p_Times] := 𝔼 /@ p;
```

```
𝔼[X⁺_{4,1} X⁺_{2,5} X⁺_{6,3}]
```

$$\mathbb{E}\left[1,\ h_4 l_1 + h_6 l_3 + h_2 l_5,\ \frac{(-1+e^{h_2})\, e_2 f_5}{h_2} + \frac{(-1+e^{h_4})\, e_4 f_1}{h_4} + \frac{(-1+e^{h_6})\, e_6 f_3}{h_6}\right]$$

```
𝕆[𝔼[ω_, L_, Q_], specs___] := 𝕆[ω e^(L+Q) /. (x : e | h)_i_ :→ ℏ x_i, specs];
```

```
Timing[$TD = 3; T1$2 = 𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3], {l1, f1, e2, l3, f3, e4, l5, f5, e6} → 1] /. h_ → h]
```

$\{0.078125, \left(1 - 2 h \hbar + h^2 \hbar^2 + \frac{2 h^3 \hbar^3}{3}\right) U[\,] + \left(3 h \hbar - 6 h^2 \hbar^2 + 3 h^3 \hbar^3\right) U[l_1] + \left(3 \hbar - \frac{3 h \hbar^2}{2} - \frac{3 h^2 \hbar^3}{2}\right) U[e_1, f_1] + \left(\frac{9 h^2 \hbar^2}{2} - 9 h^3 \hbar^3\right) U[l_1, l_1] + \left(9 h \hbar^2 - \frac{9 h^2 \hbar^3}{2}\right) U[e_1, l_1, f_1] + \frac{9}{2} h^3 \hbar^3 U[l_1, l_1, l_1] + \left(\frac{9 \hbar^2}{2} + \frac{9 h \hbar^3}{2}\right) U[e_1, e_1, f_1, f_1] + \frac{27}{2} h^2 \hbar^3 U[e_1, l_1, l_1, f_1] + \frac{27}{2} h \hbar^3 U[e_1, e_1, l_1, f_1, f_1] + \frac{9}{2} \hbar^3 U[e_1, e_1, e_1, f_1, f_1, f_1]\}$

```
T1$1 == T1$2
```

True

1. $\mathbb{O}(e^{\gamma l + \beta e} \mid le) = \mathbb{O}(e^{\gamma l + e^{\gamma} \beta e} \mid el)$,
2. $\mathbb{O}(e^{\gamma l + \beta f} \mid fl) = \mathbb{O}(e^{\gamma l + e^{\gamma} \beta f} \mid lf)$:

```
NO(x:f|e)i_ lj_[𝔼[ω_, L_, Q_]] := CF[𝔼[ω, L, e^γ α xi + (Q /. xi → 0) /. {γ → ∂lj L, α → ∂xi Q}]];
ANO(x:f|e)i_ lj_[𝔼[ω_, L_, Q_]] := CF[𝔼[ω, L, e^-γ α xi + (Q /. xi → 0) /. {γ → ∂lj L, α → ∂xi Q}]];
```

```
𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // ANOe2 l3
```

$\mathbb{E}\left[1, h_4 l_1 + h_6 l_3 + h_2 l_5, \frac{e^{-h_6} \left(-1 + e^{h_2}\right) e_2 f_5}{h_2} + \frac{\left(-1 + e^{h_4}\right) e_4 f_1}{h_4} + \frac{\left(-1 + e^{h_6}\right) e_6 f_3}{h_6}\right]$

```
𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // ANOe2 l3, {l1, f1, l3, e2, f3, e4, l5, f5, e6} → 1] ==
  𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3], {l1, f1, e2, l3, f3, e4, l5, f5, e6} → 1]
```

True

```
𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // ANOe2 l3, {l3, e2} → 1] == 𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3], {e2, l3} → 1]
```

True

```
(𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // ANOe2 l3 // NOe2 l3) == 𝔼[X⁺4,1 X⁺2,5 X⁺6,3]
```

True

3. $\mathbb{O}(e^{\beta e + \alpha f + \delta ef} \mid fe) = \mathbb{O}(\nu e^{\nu(-\alpha\beta h + \beta e + \alpha f + \delta ef)} \mid ef)$, with $\nu = (1 + h\delta)^{-1}$:

```
NOfi_ ej_→k_[𝔼[ω_, L_, Q_]] := CF[
    𝔼[ν ω, L, ν (-α β hk + β ek + α fk + δ ek fk) + (Q /. fi | ej → 0)]
      /. ν → (1 + hk δ)^-1 /. {α → ∂fi Q /. ej → 0, β → ∂ej Q /. fi → 0, δ → ∂fi,ej Q}
  ];
```

```
𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // NOf3 e4→7
```

$\mathbb{E}\left[1, h_4 l_1 + h_6 l_3 + h_2 l_5, \frac{1}{h_2 h_4 h_6} \left(\left(\left(-1 + e^{h_4}\right) e_7 f_1 h_2 + \left(-1 + e^{h_2}\right) e_2 f_5 h_4\right) h_6 + \left(-1 + e^{h_6}\right) e_6 h_2 \left(f_7 h_4 - \left(-1 + e^{h_4}\right) f_1 h_7\right)\right)\right]$

```
𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // NOf3 e4→7, {l1, f1, e2, l3, e7, f7, l5, f5, e6} → 1] ==
   𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3], {l1, f1, e2, l3, f3, e4, l5, f5, e6} → 1] /. h_ → h
```

True

The Stitching Formula

```
mi_,j_→k_[Z_] := Module[{x, z}, CF[(Z // NOfi ej→x // NOli ex // NOfx lj) /. z_i|j|x → zk]]
```

```
𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // m1,2→1
```

$$\mathbb{E}\left[1,\ h_4 l_1 + h_6 l_3 + h_1 l_5,\ \frac{1}{h_1 h_4 h_6}\left(\left(-1+e^{h_6}\right) e_6 f_3 h_1 h_4 + \left(\left(-1+e^{h_4}\right) e_4 \left(f_1 - \left(-1+e^{h_1}\right) f_5\right) h_1 + e^{h_4}\left(-1+e^{h_1}\right) e_1 f_5 h_4\right) h_6\right)\right]$$

```
𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // m1,2→1, {e1, l1, f1, l3, f3, e4, l5, f5, e6} → 1] ==
  𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3], {l1, f1, e2, l3, f3, e4, l5, f5, e6} → 1] /. h_ → h
```

True

```
𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // m1,2→1 // m1,3→1
```

$$\mathbb{E}\left[1,\ h_4 l_1 + h_6 l_1 + h_1 l_5,\ \frac{1}{h_1 h_4 h_6}\left(\left(-1+e^{h_6}\right) e_6 f_1 h_1 h_4 + \left(\left(-1+e^{h_4}\right) e_4 \left(e^{h_6} f_1 - \left(-1+e^{h_1}\right) f_5\right) h_1 + e^{h_4}\left(-1+e^{h_1}\right) e_1 f_5 h_4\right) h_6\right)\right]$$

```
𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // m1,2→1 // m1,3→1, {e1, l1, f1, e4, l5, f5, e6} → 1] ==
  𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3], {l1, f1, e2, l3, f3, e4, l5, f5, e6} → 1] /. h_ → h
```

True

```
𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // m1,2→1 // m1,3→1 // m1,4→1
```

$$\mathbb{E}\left[\frac{1}{1+e^{h_6}\left(-1+e^{h_1}\right)},\ h_6 l_1 + h_1\left(l_1 + l_5\right),\ \left(\left(-1+e^{h_6}\right) e_6 \left(f_1 + \left(-1+e^{h_1}\right)^2 f_5\right) h_1 + e^{h_1}\left(-1+e^{h_1}\right) e_1 \left(e^{2 h_6} f_1 + f_5\right) h_6\right) \Big/ \left(\left(1 - e^{h_6} + e^{h_1+h_6}\right) h_1 h_6\right)\right]$$

```
𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // m1,2→1 // m1,3→1 // m1,4→1, {e1, l1, f1, l5, f5, e6} → 1] ==
  𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3], {l1, f1, e2, l3, f3, e4, l5, f5, e6} → 1] /. h_ → h
```

True

```
𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // m1,2→1 // m1,3→1 // m1,4→1 // m1,5→1 // m1,6→1
```

$$\mathbb{E}\left[\frac{e^{-h_1}}{1 - e^{h_1} + e^{2 h_1}},\ 3 h_1 l_1,\ \frac{\left(-1+e^{3 h_1}\right) e_1 f_1}{h_1}\right]$$

```
𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3] // m1,2→1 // m1,3→1 // m1,4→1 // m1,5→1 // m1,6→1, {e1, l1, f1} → 1] ==
  𝕆[𝔼[X⁺4,1 X⁺2,5 X⁺6,3], {l1, f1, e2, l3, f3, e4, l5, f5, e6} → 1] /. h_ → h
```

True

The Main $\mathfrak{g}_0$ Theorem, Polished Version

Polished Version. With $\overline{e} = \frac{(e^h-1)}{h} e$, the $\mathfrak{g}_0$ invariant of any S-component tangle T can be written in the form $Z(T) = \mathbb{O}\left(\omega^{-1} e^{L+\omega^{-1} Q} \mid \prod_{i\in S} \overline{e}_i l_i f_i\right)$, where ω is a scalar (meaning, a polynomial in the variables $t_i = e^{h_i}$), where $L = \sum a_{ij} h_i l_j$ is a balanced quadratic in the variables h_i and l_j with integer coefficients a_{ij} and where $Q = \sum b_{ij} \overline{e}_i f_j$ is a balanced quadratic in the variables $\overline{e}_i$ and f_j with scalar coefficients b_{ij}. Furthermore, after setting $t_i = t$ for all i, the invariant $Z(T)$ is poly-time computable.

Debt. Implement (and verify!).

Just a Computation

```
With[{ψ = ν e^(ν (t e f-α β h+α f+β e)) /. ν → (1 + t h)^-1}, Simplify@{∂t ψ - ∂α,β ψ, ψ /. t → 0}]
```

$$\left\{0,\ e^{f\alpha + e\beta - h\alpha\beta}\right\}$$